

The House Warming Model.

Michael Monagan
 Department of Mathematics
 Simon Fraser University
 mmonagan@cecm.sfu.ca


Maple commands: solve, collect, dsolve, odeplot, DEplot

In the Fall of 2012 and 2013 I taught MACM 204 Computing and Calculus to second and third year students. The course is intended to introduce our mathematics majors at SFU to a general purpose mathematical software package. We use Maple. Students taking the course also include students from economics, the sciences and engineering. They have taken first year differential and integral calculus and a first programming course. One of the main topics in the course is modeling with differential equations. Unlike physics, chemistry, biology and engineering students, mathematics, statistics and computing science students rarely study physical systems. So we try in this course to not only teach students how to use Maple to solve a variety of problems, but how to model physical systems with DEs. In this article and a subsequent article I will give two non-standard examples that I've found helpful. The one presented here is a first order linear system. Second order systems and non-linear systems are too complicated to study analytically in a first course. I also like this example because it's flexible. And, at least for now, students won't find the solutions to it online! They'll find a very different kind of house warming!!

Suppose we have a house with two rooms A and B (see below) with a furnace (a heater) F in room B. We want to model how the two rooms heat up when the heater is switched on. Let

$A(t)$ be the temperature in room A at time t ,
 $B(t)$ be the temperature in room B at time t ,
 A_m be the outside air temperature (assumed constant), and
 F be the heat generated by the furnace (assumed constant rate).

Here is a diagram showing how heat flows between the rooms and to the outside.



$$A'(t) = k_3(A - A_m) - k_1(B - A)$$

$$B'(t) = k_1(B - A) - k_2(B - A_m) + F$$

I drew the sketch in Maple using the "canvas" facility. My sketch has used actual text and

also hand drawn text, actual components (the rectangles) and hand drawn arrows (there are actual arrows if you want "nice" arrows). You can also copy and paste mathematics and other objects onto the canvas. I typed the DEs into the canvas from inside a text box.

How did I get those two differential equations?

Newton's law of cooling says that the rate at which a body cools is proportional to the difference between the temperature of the body and the temperature of the surrounding medium. Here, in Room A, the temperature of room A is $A(t)$ and the outside temperature is A_m (assumed constant). So heat flows from Room A to the outside through the walls and room at the rate

$$k_3 \cdot (A(t) - A_m)$$

where k_3 is some constant. The larger k_3 the faster heat moves from room A to the outside.

We can use the parameter k_3 to simulate a house with no insulation and a house with good insulation. Now heat also flows from room B to room A through the wall and the door connecting the two rooms. I guess if the door is open then heat will flow faster from room B to room A! This is specified by k_1 . Hence we have

```
> de1 := diff(A(t),t) = +k1*(B(t)-A(t)) -k3*(A(t)-Am);
```

$$de1 := \frac{d}{dt} A(t) = k_1 (B(t) - A(t)) - k_3 (A(t) - A_m)$$

Repeating this for room B, and including the heat from the furnace F which we assume is constant we have

```
> de2 := diff(B(t),t) = -k1*(B(t)-A(t))-k2*(B(t)-Am) + F;
```

$$de2 := \frac{d}{dt} B(t) = -k_1 (B(t) - A(t)) - k_2 (B(t) - A_m) + F$$

Temperature equilibrium.

So what happens when say the heater F is switched on? Assuming $A(0) = A_m$ and $B(0) = B_m$, obviously room A heats up but also room B heats up. And they heat up until the heat from the furnace equals the heat flowing from room A and B to the outside. I.e., we will get to some equilibrium temperature at which there is no change in temperature, that is

$$A'(t) = 0 \text{ and } B'(t) = 0 .$$

We can find this equilibrium temperature without solving the differential equations. Instead, we solve the two equations algebraically in Maple.

```
> sys := { rhs(de1)=0, rhs(de2)=0 };
```

$$sys := \{ k_1 (B(t) - A(t)) - k_3 (A(t) - A_m) = 0, -k_1 (B(t) - A(t)) - k_2 (B(t) - A_m) + F = 0 \}$$

```
> TempEquil := solve( sys, {A(t),B(t)} );
```

$$TempEquil := \left\{ A(t) = \frac{A_m k_1 k_2 + A_m k_1 k_3 + A_m k_2 k_3 + F k_1}{k_1 k_2 + k_1 k_3 + k_2 k_3}, B(t) = \frac{A_m k_1 k_2 + A_m k_1 k_3 + A_m k_2 k_3 + F k_1 + F k_3}{k_1 k_2 + k_1 k_3 + k_2 k_3} \right\}$$

Looking at these solutions, I can't really see what they mean. Can we express the solutions in the form $A_m + f(k_1, k_2, k_3, F)$. The **collect** command collects (or groups) terms in one or

more variables together. Let's collect in A_m

```
> collect( TempEquil, Am );
```

$$\left\{ A(t) = A_m + \frac{Fk_1}{k_1 k_2 + k_1 k_3 + k_2 k_3}, B(t) = A_m + \frac{Fk_1 + Fk_3}{k_1 k_2 + k_1 k_3 + k_2 k_3} \right\}$$

That's much better. We can see that the equilibrium temperatures are of the form $A_m + f(k_1, k_2, k_3) \cdot F$. By specifying a third argument we can get Maple to simplify each term in A_m

```
> collect( TempEquil, Am, simplify );
```

$$\left\{ A(t) = A_m + \frac{Fk_1}{k_1 k_2 + k_1 k_3 + k_2 k_3}, B(t) = A_m + \frac{F(k_1 + k_3)}{k_1 k_2 + k_1 k_3 + k_2 k_3} \right\}$$

That's a very nice result. Now that collect command may seem mysterious. Consider a polynomial

```
> poly := a[0]+a[1]*x+a[2]*x^2+b[1]*x;
```

$$poly := x^2 a_2 + x a_1 + x b_1 + a_0$$

```
> collect(poly,x,G);
```

$$G(a_2) x^2 + G(a_1 + b_1) x + G(a_0)$$

Now you can see what collect does. It groups the terms in powers of x together and applies the function G to the coefficients.

Back to the house warming problem. My statement "until the heat from the furnace equals the heat flowing from room A and B to the outside" is really a conservation law stating that when the room temperatures are not changing, the heat coming into the house from the heater F must balance the heat going out of the house. We shall check this explicitly

```
> eqn := F = k3*(A(t)-Am) + k2*(B(t)-Am);
```

$$eqn := F = k_3 (A(t) - A_m) + k_2 (B(t) - A_m)$$

```
> simplify ( subs(TempEquil,eqn) );
```

$$F = F$$

Analytical Solutions

Now we use the **dsolve** command to solve the DEs for $A(t)$ and $B(t)$ for initial values $A(0) = A_m$ and $B(0) = A_m$. The solutions are ugly. We can get modestly compact formulae for the special case $k_2 = k_3$, that is, for identical rooms.

```
> k2 := k3;
```

```
sols := dsolve( {de1,de2,A(0)=Am,B(0)=Am}, {A(t),B(t)} );
```

$$k_2 := k_3$$

$$\begin{aligned} sols := & \left\{ A(t) = -\frac{1}{2} \frac{e^{-k_3 t} F}{k_3} + \frac{1}{2} \frac{e^{(-2k_1 - k_3)t} F}{2k_1 + k_3} + \frac{2A_m k_1 k_3 + A_m k_3^2 + Fk_1}{k_3 (2k_1 + k_3)}, B(t) \right. \\ & = \frac{1}{k_3 (2k_1 + k_3)} \left(-\frac{e^{-(2k_1 + k_3)t} Fk_3 k_1}{2k_1 + k_3} - \frac{1}{2} \frac{e^{-(2k_1 + k_3)t} Fk_3^2}{2k_1 + k_3} - e^{-k_3 t} Fk_1 \right. \\ & \left. \left. - \frac{1}{2} e^{-k_3 t} Fk_3 + 2A_m k_1 k_3 + A_m k_3^2 + Fk_1 + Fk_3 \right) \right\} \end{aligned}$$

I'd like to pause here and illustrate again the **collect** command. We already know that the temperature equilibrium point can be expressed as $A_m + f(k_1, k_2, k_3) \cdot F$. So we expect that

the solutions for $A(t)$ and $B(t)$ can also be expressed in this way. Hence

$$\begin{aligned} &> \text{collect(sols, Am, simplify);} \\ &\left\{ A(t) = Am + \frac{1}{2} \frac{F(e^{-(2k1+k3)t} k3 - 2e^{-k3t} k1 - k3e^{-k3t} + 2k1)}{k3(2k1+k3)}, B(t) = Am \right. \\ &\quad \left. - \frac{1}{2} \frac{(e^{-(2k1+k3)t} k3 + 2e^{-k3t} k1 + k3e^{-k3t} - 2k1 - 2k3) F}{k3(2k1+k3)} \right\} \end{aligned}$$

Can we further simplify the solutions? Let us also collect on the exponentials. We could input the actual exponentials (there are two) separately but collect allows us to specify all exponentials by using just **exp** as follows.

$$\begin{aligned} &> \text{collect(sols, [Am,exp], simplify);} \\ &\left\{ A(t) = Am - \frac{1}{2} \frac{e^{-k3t} F}{k3} + \frac{1}{2} \frac{e^{-(2k1+k3)t} F}{2k1+k3} + \frac{Fk1}{k3(2k1+k3)}, B(t) = Am - \frac{1}{2} \frac{e^{-k3t} F}{k3} \right. \\ &\quad \left. - \frac{1}{2} \frac{e^{-(2k1+k3)t} F}{2k1+k3} + \frac{F(k1+k3)}{k3(2k1+k3)} \right\} \end{aligned}$$

Graphing Solutions of DEs

Maple has three facilities for generating plots of solutions of systems of DEs. The first is to find exact formulas for the solutions using **dsolve** then graph them using the plot command. The other two facilities are the **DEplot** facility in the DEtools package and the **odeplot** command in the plots package, both use numerical methods to solve the system.

I'll illustrate all three. First we have to fix numerical values for the parameters.

```
> Am := 0;
k2 := k3;
k1 := 2*k2;
k3 := 0.1;
F := 5;
```


```
Am:= 0
k2:= k3
k1:= 2 k3
k3:= 0.1
F:= 5
```

```
> sol := dsolve( {de1,de2,A(0)=Am,B(0)=Am}, {A(t),B(t)} );
sol:= {A(t) = -25 e^{-1/10 t} + 5 e^{-1/2 t} + 20, B(t) = -25 e^{-1/10 t} - 5 e^{-1/2 t} + 30}
```

The solutions returned by dsolve are equations. We want to graph the right-hand-sides of the equations. We can select them and use copy and paste or extract them using


```
> map( rhs, sol );
{-25 e^{-1/10 t} - 5 e^{-1/2 t} + 30, -25 e^{-1/10 t} + 5 e^{-1/2 t} + 20}
```

```
> plot( map(rhs,sol), t=0..50 );
```



The **DEplot** command in the DEtools package generates a phase-portrait plot. That is, a field plot together with plots of solution curves for given initial values. I'm going to use two initial values so there are two solution curves (one in blue, the other in green). The nicest options for the field plot are `dirfield=n` which generates `n` randomly placed arrows to show the direction field. For the arrows I've used comets. Each comet is curved showing the direction of the field at that point.

```
> with(DEtools):
DEplot( {de1,de2}, {A(t),B(t)}, t=0..30, A=0..50,B=0..50,
[[A(0)=Am,B(0)=Am],[A(0)=50,B(0)=30]], dirfield=100, linecolor=[blue,
green], arrows=comet);
```



The other way to get a plot is to first solve the DEs numerically.

```
> Temp := dsolve( {de1,de2,A(0)=Am,B(0)=Am}, {A(t),B(t)}, numeric );
Temp:=proc(x_rkf45) ... end proc
```

The output is a Maple procedure. The name `rkf45` is telling us the name of the method that will be used. It is the well known Runge-Kutta Fehlberg 4-5 method for solving differential equations. But how do we use that output? The output procedure should be thought as $Temp(t)$ that is on input of t it outputs $A(t)$ and $B(t)$. E.g.

```
> Temp(1.0);  
[t = 1.0, A(t) = 0.411717833437737, B(t) = 4.34641126477970]
```

```
> Temp(10.0);  
[t = 10.0, A(t) = 10.8367020049238, B(t) = 20.7693260449041]
```

```
> Temp(100.0);  
[t = 100.0, A(t) = 19.9988674398753, B(t) = 29.9988674398757]
```


```
> Temp(1000.0);  
[t = 1000.0, A(t) = 19.9999972539894, B(t) = 30.0000027460106]
```

Okay, so we can see that, numerically, $\lim_{t \rightarrow \infty} A(t) = 20$ and $\lim_{t \rightarrow \infty} B(t) = 30$. Let's check that


```
> TempEquil;  
{A(t) = 20.00000000, B(t) = 30.00000000}
```

Well that's good. I guess that means the numerical method is working correctly. So how do we get a plot of the solutions? The odeplot command in the plottools package allows us to do this as follows. There are two examples, first is to plot $A(t)$ and $B(t)$ verses t . The second plots $A(t)$ verses $B(t)$ as in the phase portrait plot above.

```
> plots[odeplot]( Temp, [[t,A(t)],[t,B(t)]], 0..50, legend=[A(t),B(t)] )  
;
```




```
> plots[odeplot]( Temp, [A(t),B(t)], 0..50 );
```



What to do with this model? One of the difficult aspects of teaching differential equations

is teaching the students how to build models. Most of the time we just give them the differential equations and tell them what the terms mean. We would do that, for example, with the Lotka-Volterra predator prey system. We rarely ask the student to come up with the model. One thing we can do with this house warming model is ask students to do this experiment with a different house. Here are two houses that I have tried.



I have noticed that many students make the following error when inputting a differential equation. Both in 1D Maple input and 2D Maple input. Instead of inputting, for example

```
> deB := -k1*(B(t)-Am) - k2*(B(t)-A(t)) - k3*(B(t)-C(t));
      deB:= -k1 (B(t) - Am) - k2 (B(t) - A(t)) - k3 (B(t) - C(t))
```

they input

```
> deB := -k1(B(t)-Am) - k2*(B(t)-A(t))-k3(B(t)-C(t));
      deB:= -k1(B(t) - Am) - k2 (B(t) - A(t)) - k3(B(t) - C(t))
```

Here Maple treats k1 and k3 as functions and it cannot do anything useful from this point on. It's an easy mistake to make. It's worthwhile deliberately making this mistake when giving a demo and asking students to find the error. It'll save some grief later.